Волновая функция для 4-солитонного решения КдФ

К лекции 3 (2024)

Формула для n-солитонного решения КдФ:

(1) $u = -2\,\partial_x^2 \log W(y_1, \ldots, y_n), \quad y_j = e^{X_j} + (-1)^{n-j} e^{-X_j}, \quad X_j = k_j\, x + 4\, k_j^3\, t + \delta_j, \quad 0 < k_n < \ldots < k_1.$

(2) $c_j = (-1)^{n-j} e^{2\,\delta_j}, \quad j = 1, \ldots, n.$

Соответствующее частное решение уравнения Шрёдингера

(3) $\psi'' = (u - \lambda)\,\psi, \quad -z^2 = \lambda,$

имеет вид

(4) $\psi(z) = e^X\, \Phi(z) = e^{z\,x + 4\,z^3\,t} \left(z^n + z^{n-1}\,\varphi_1 + \ldots + \varphi_n\right) = W\!\left(y_1, \ldots, y_n, e^{z\,x + 4\,z^3\,t}\right) \Big/ W(y_1, \ldots, y_n).$

Мы ограничимся случаем $n = 4$ для конкретных численных параметров k_j, δ_j.

• Одна из функций $\psi(z)$ и $\psi(-z)$ экспоненциально убывает при $x \to +\infty$, другая при $x \to -\infty$. В точках $\lambda_j = -k_j^2$ эти функции становятся линейно зависимыми и определяют собственную функцию оператора Шрёдингера, с точностью до нормировки.

• Собственные значения принято нумеровать по возрастанию.У нас они все отрицательные, $\lambda_j = -k_j^2$, поэтому λ_n – самое маленькое по абсолютной величине:

$\lambda_1 < \ldots < \lambda_n < 0.$

Числа k_j мы для определенности будем считать положительными, тогда они упорядочены так, как указано в (1).

• Правило чередования знаков (2) обеспечивает регулярность потенциала (и волновых функций).

1 Вычисления

In[1]:=
```
n = 4;  (* число солитонов *)
X[j_] := x K[j] + 4 t K[j]^3 + d[j]
Y[j_] := Exp[X[j]] + (-1)^(n-j) Exp[-X[j]]

w = Det[Table[D[Y[j], {x, k}], {j, 1, n}, {k, 0, n - 1}]];
w1 = Det[Append[Table[D[Y[j], {x, k}], {j, 1, n}, {k, 0, n}], Table[z^k, {k, 0, n}]]];

(* параметры *)
par = {K[1] → 0.95, K[2] → 0.75, K[3] → 0.6, K[4] → 0.35,
   d[1] → -5, d[2] → 1.3, d[3] → 0.3, d[4] → -0.2};

(* потенциал *)
u = -2 D[D[w, x] / w, x] /. par;
cu = Compile[{{x, _Real}, {t, _Real}}, Evaluate[u]];

(* psi-функция. Множитель 3 введен произвольно,
чтобы отмасштабировать график в данном конкретном примере *)
Psi = 3 Exp[x z + 4 t z^3] w1 / w /. par;
cpsi = Compile[{{x, _Real}, {t, _Real}, {z, _Real}}, Evaluate[Psi]];

(* регуляризуем собственные функции, чтобы подавить численные ошибки на краях *)
regpsi[x_, t_, z_] := If[
  Min[Abs[Table[K[j] /. par, {j, 1, n}] - Abs[z]]] > 0.001,
  cpsi[x, t, z],
  If[x z + 4 t z^3 < 0, cpsi[x, t, z], cpsi[0, 0, z] / cpsi[0, 0, -z] × cpsi[x, t, -z]]]
```

2 Зависимость от *z* при фиксированном *t*

Вот на что нужно обратить внимание:

– при $z \geq k_1$ функция $\psi(z)$ не имеет нулей на оси x;

– при уменьшении z до 0, нули «заходят» в $\psi(z)$ через $x = +\infty$, при прохождении z через k_j, и движутся налево. В результате, при каждом $z \in \left[k_{j+1}, k_j\right)$ эта функция имеет j нулей по x (n при $0 \leq z < k_n$);

– при дальнейшем уменьшении z нули «выходят» через $x = -\infty$, когда z проходит $-k_j$;

– у функции $\psi(-z)$ естественно, все наоборот.

– при каждом фиксированном x функция $\psi(z, x)$ имеет ровно n нулей по z:

$$\psi(z) = e^X \Phi(z) = e^X (z - r_1(x)) \ldots (z - r_n(x)),$$

In[14]:=

```
grz[z_, t_] := Plot[{regpsi[x, t, z], regpsi[x, t, -z], cu[x, t]}, {x, -40, 40},
  PlotRange → {{-40, 40}, {-2.1, 2.1}},
  Filling → {3 → 0},
  AxesLabel → {" x", "ψ(x,t;z) "},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  PlotStyle → {Blue, Red, Black},
  ImageSize → 500,
  Epilog → {
    Text["t = " <> ToString[N[t]], {13, 2.0}, {-1, 0}],
    Style[Text["z = " <> ToString[N[z]], {13, 1.7}, {-1, 0}],
     If[Min[Abs[Abs[z] - {K[1], K[2], K[3], K[4], 0} /. par]] < 0.001,
      {Bold, Red}, Black]]},
  PlotLegends → LineLegend[Automatic, {"ψ(z)", "ψ(-z)", "u"}]
 ]

Manipulate[grz[z, t],
 {{z, 1}, -1, 1, 0.01, Appearance → {"Labeled", "Open"}},
 {{t, 0.3}, -8, 8, 0.01, Appearance → "Labeled"}]
```

Out[15]=



3 Зависимость от *t* для собственных функций

Нормируем с.ф. на 1.

In[ ]:=

```
Do[norm[i] = Sqrt[NIntegrate[regpsi[x, 0, K[i] /. par]^2, {x, -40, 40}]], {i, 1, 4}]

grt[t_] := Plot[{
    regpsi[x, t, K[1] /. par] / norm[1] - K[1]^2 /. par,
    regpsi[x, t, K[2] /. par] / norm[2] - K[2]^2 /. par,
    regpsi[x, t, K[3] /. par] / norm[3] - K[3]^2 /. par,
    regpsi[x, t, K[4] /. par] / norm[4] - K[4]^2 /. par,
    cu[x, t]}, {x, -41, 41},
  PlotRange → {{-41, 41}, {-2.05, 0.35}},
  Filling → {1 → (-K[1]^2 /. par),
    2 → (-K[2]^2 /. par),
    3 → (-K[3]^2 /. par),
    4 → (-K[4]^2 /. par),
    5 → 0},
  AxesLabel → {" x", None},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  PlotStyle → {Red, Green, Blue, Cyan, Black},
  ImageSize → 500,
  Epilog → {
    Text["t = " <> ToString[N[t]], {30, 0.2}, {-1, 0}]},
  PlotLegends → LineLegend[Automatic, {"ψ(k₁)", "ψ(k₂)", "ψ(k₃)", "ψ(k₄)", "u"}]
 ]

Manipulate[grt[t],
 {{t, -12}, -12, 12, 0.05, Appearance → {"Labeled", "Open"}}]
```

Out[ ]=